# プログラムのページ

担当 吉沢　正

## 7103 奇関数の最良近似有理式を求めるプログラム

山下真一郎（富士通株式会社）

ここに述べるプログラムは，重率を考慮した多項式を含む有理式によって，奇関数の最良近似式を求めるもので，相対誤差，または絶対誤差を最小にする多項式，または有理式の近似式を求めることができる．

計算法を説明するのに必要な諸定義を，次のようにおく．

被近似関数　$F(X)$ (1)

近似区間　$A \leqq X \leqq B$，ただし $A=-B$ (2)

近似式　$P_L(X)/Q_M(X)$

$$=\sum_{i=0}^{L} p_i X^{2i+1} \Big/ \left(1.0+\sum_{j=1}^{M} q_j X^{2j}\right) \tag{3}$$

重率関数　$W(X)$，$W(A \leqq X \leqq B) \neq 0$ (4)

誤差関数　$E(X)=W(X)\{F(X)-P_L(X)/Q_M(X)\}$ (5)

近似式の次数　$N=2L+2M+2$ (6)

被近似式が奇関数であるから，分子の多項式は奇数次の項だけから成り，分母の多項式は偶数次の項だけから成る近似式を想定する．近似式の次数は分母子の次数をそれぞれ次の項まで想定すれば，$2L+2M+3$ のはずであるが，$X=0$ の点で $E(X)=0$ とならず，これでは具合が悪い．

最良近似式は近似式を逐次補正して求める．その初期近似式は，$E(X)$ の0点の近似値での有理式補間法で求める．

0点の近似値：

$$X_i=\left(\frac{A-B}{2}\right)\cos\left(\frac{2i+1}{N+1}\cdot\frac{\pi}{2}\right)+\left(\frac{B+A}{2}\right) \tag{7}$$

$i=0,\ 1,\ 2,\ \cdots,\ L+M$

有理補間法というのは，$F(X)$ と $P_L(X)/Q_M(X)$ が一致するような，$P_L(X)$，$Q_M(X)$ を求めることである．すなわち

$$\left.\begin{aligned} &F(X_i)-P_L(X_i)/Q_M(X_i)=0\\ &\therefore\ P_L(X_i)-F(X_i)\{Q_M(X_i)-1.0\}=F(X_i)\\ &\therefore\ \sum_{k=0}^{L}(X_i^{2k+1})p_k+\sum_{k=1}^{M}(-F(X_i)X_i^{2k})q_k\\ &\qquad =F(X_i)\end{aligned}\right\} \tag{8}$$

$i=0,\ 1,\ 2,\ \cdots,\ L+M$

これは，$(L+M+1)$ 元連立一次方程式である．

逐次補正は次のようにして行なう．何回目かの近似式を $P_L(X)/Q_M(X)$ とし，最良近似式を $P_L^*(X)/Q_M^*(X)$ とする．この誤差関数をそれぞれ次のように表わす．

$$\left.\begin{aligned} E(X)&=W(X)\{F(X)-P_L(X)/Q_M(X)\}\\ E^*(X)&=W(X)\{F(X)-P_L^*(X)/Q_M^*(X)\}\end{aligned}\right\} \tag{9}$$

$P_L(X)$，$Q_M(X)$ を $\Delta P_L(X)$，$\Delta Q_M(X)$ 補正すれば，$P_L^*(X)$，$Q_M^*(X)$ が得られるとする．すなわち

$$\left.\begin{aligned} P_L^*(X)&=P_L(X)+\Delta P_L(X)\\ Q_M^*(X)&=Q_M(X)+\Delta Q_M(X)\end{aligned}\right\} \tag{10}$$

$E(X)$ の極大点を $X_j$ とし，$E^*(X)$ の極大点とあまり変わらないとすれば，次式を得る．

$$\left.\begin{aligned} &E^*(X_j)=(-1)^j\rho\\ &\rho=\text{定数},\ j=0,\ 1,\ 2,\ \cdots L+M+1\end{aligned}\right\} \tag{11}$$

このとき

$$\begin{aligned} &E(X_j)-E^*(X_j)\\ &=W(X_j)\frac{\Delta P_L(X_j)Q_M(X_j)-\Delta Q_M(X_j)P_L(X_j)}{Q_M(X_j)\{Q_M(X_j)+\Delta Q_M(X_j)\}}\\ &=E(X_j)-(-1)^j\rho \end{aligned} \tag{12}$$

$Q_M(X_j)$ に比べて，$\Delta Q_M(X_j)$ は十分小さいと仮定して，(12)式の分母の $\Delta Q_M(X_j)$ を無視すると，(12)式は $\Delta P_L(X)$，$\Delta Q_M(X)$ および $\rho$ に関する $(L+M+2)$ 元連立一次方程式となる．

$$\Delta P_L(X)=\sum_{i=0}^{L}\Delta p_k X^{2i+1},\quad \Delta Q_M(X)=\sum_{i=1}^{M}\Delta q_i X^{2i}$$

とおき，(12)式の相隣る式を加え，$\rho$ を消去すると，つぎの $(L+M+1)$ 元連立一次方程式を得る．

$$\begin{aligned} &\sum_{k=0}^{L}\left\{\frac{W(X_j)X_j^{2k+1}}{Q_M(X_j)}+\frac{W(X_{j+1})X_{j+1}^{2k+1}}{Q_M(X_{j+1})}\right\}\Delta p_k\\ &\quad+\sum_{k=1}^{M}\left\{-\frac{W(X_j)P_L(X_j)X_j^{2k}}{Q_M^2(X_j)}\right.\\ &\quad\left.-\frac{W(X_{j+1})P_L(X_{j+1})X_{j+1}^{2k}}{Q_M^2(X_{j+1})}\right\}\Delta q_k\\ &\quad=E(X_j)+E(X_{j+1}) \end{aligned} \tag{13}$$

この連立方程式を解いて，$\Delta P_L(X)$，$\Delta Q_M(X)$ を

得て，(10) 式で $P_L(X)$，$Q_M(X)$ を補正する.

極大点 $X_j$ は次の近似値を初期値にして，山登り法で求める.

極大点の近似値：

$$X_j=\left(\frac{B-A}{2}\right)\cos\left(\frac{2j}{N+1}\cdot\frac{\pi}{2}\right)+\left(\frac{B+A}{2}\right) \qquad (14)$$

$$j=0,\ 1,\ 2,\ \cdots,\ L+M+1$$

プログラムは FACOM 230-60 FORTRAN で作った.

主プログラムでは，近似式の次数，関数，重率，誤差，収束判定用定数を定義している．ほとんどの計算は KIGU で行なう．POL は多項式の値を求め，PQ は有理式の値を求める． YAMA は極大点を求め，CURVE はグラフを描く．主プログラムおよび結果は sin の絶対誤差を最小にするものである．紙数の関係で結果は一部を示した.

このプログラムを他の計算機用に記述するためにはつぎの点に注意して書き換えが必要である.

1. 添字式が式になっている.
2. DO のパラメータが式になっている.
3. 書式の文字表現が '文字列' となっている.
4. 書式の欄区切りカンマ“，”が省略されている.
5. 多重代入文が使われている.
6. 混合演算が使われている.
7. 印刷の 1 行が 136 文字である (グラフのところだけ 100 字をこえている.).
8. 数値は最大約 $10^{+77}$ まで表現できる.
9. 倍精度実数は約 18 けたの精度がある.

このように，書き換えの必要な文には，文の内部番号に丸印をつけておいた.

新しい関数について，近似式を求めるときは，主プログラムと被近似関数を作り変えればよい．すなわち，つぎのものを用意すればよい.

1. 被近似関数 $F(X)$
2. 重率関数 $W(X)$
3. 誤差関数 $E(X)$
4. 近似区間の両端 $A$, $B$,
5. 近似式の次数 $L$, $M$, $N$; $N=L+M$
6. 収束判定用定数 $EPS_1$, $EPS_2$

$\sin\left(\frac{\pi}{2}X\right)$，$|X|\leqq 1.0$ の分子が奇数次多項式，分母が偶数次多項式となるような近似式の次数と誤差の関係は，**第 1 表**のようになっている.

この表から，係数の個数が同じ場合には，分子の次数が分母より大きいほうが有利である．しかし，これは計算誤差の問題とは別である．分子の次数が 1 であると，分母の係数はすべて正となり，けた落ちがない.

けた落ちするといっても，最初のけたから消失するわけではないので，sin の計算は，多項式で計算するのが，ほぼ最良の計算法ということができよう.

第 1 表 $\sin\left(\frac{\pi}{2}X\right)$ の近似式の次数と誤差の関係

| 分子＼分母 | 0 | 2 | 4 | 6 | 8 |
|---|---|---|---|---|---|
| 1 | | $1.2\times10^{-2}$ | $8.9\times10^{-4}$ | $6.4\times10^{-5}$ | $4.6\times10^{-6}$ |
| 3 | $4.5\times10^{-3}$ | $7.8\times10^{-5}$ | $1.4\times10^{-6}$ | $2.5\times10^{-8}$ | |
| 5 | $6.8\times10^{-5}$ | $4.3\times10^{-7}$ | $3.3\times10^{-9}$ | | |
| 7 | $5.9\times10^{-7}$ | $1.8\times10^{-9}$ | | | |
| 9 | $3.3\times10^{-9}$ | | | | |

第2表　第4表の結果の誤差グラフ

| X | E |
|---|---|
| -1.00000 | 0.334E-08 |
| -0.98000 | -0.190E-08 |
| -0.96000 | -0.333E-08 |
| -0.94000 | -0.267E-08 |
| -0.92000 | -0.109E-08 |
| -0.90000 | 0.641E-09 |
| -0.88000 | 0.208E-08 |
| -0.86000 | 0.300E-08 |
| -0.84000 | 0.334E-08 |
| -0.82000 | 0.313E-08 |
| -0.80000 | 0.249E-08 |
| -0.78000 | 0.156E-08 |
| -0.76000 | 0.472E-09 |
| -0.74000 | -0.631E-09 |
| -0.72000 | -0.163E-08 |
| -0.70000 | -0.245E-08 |
| -0.68000 | -0.302E-08 |
| -0.66000 | -0.330E-08 |
| -0.64000 | -0.330E-08 |
| -0.62000 | -0.304E-08 |
| -0.60000 | -0.254E-08 |
| -0.58000 | -0.186E-08 |
| -0.56000 | -0.105E-08 |
| -0.54000 | -0.188E-09 |
| -0.52000 | 0.678E-09 |
| -0.50000 | 0.149E-08 |
| -0.48000 | 0.219E-08 |
| -0.46000 | 0.275E-08 |
| -0.44000 | 0.313E-08 |
| -0.42000 | 0.332E-08 |
| -0.40000 | 0.331E-08 |
| -0.38000 | 0.311E-08 |
| -0.36000 | 0.273E-08 |
| -0.34000 | 0.220E-08 |
| -0.32000 | 0.156E-08 |
| -0.30000 | 0.828E-09 |
| -0.28000 | 0.616E-10 |
| -0.26000 | -0.704E-09 |
| -0.24000 | -0.143E-08 |
| -0.22000 | -0.206E-08 |
| -0.20000 | -0.261E-08 |
| -0.18000 | -0.301E-08 |
| -0.16000 | -0.326E-08 |
| -0.14000 | -0.334E-08 |
| -0.12000 | -0.325E-08 |
| -0.10000 | -0.300E-08 |
| -0.08000 | -0.260E-08 |
| -0.06000 | -0.207E-08 |
| -0.04000 | -0.144E-08 |
| -0.02000 | -0.738E-09 |
| 0.0 | 0.0 |
| 0.02000 | 0.738E-09 |
| 0.04000 | 0.144E-08 |
| 0.06000 | 0.207E-08 |
| 0.08000 | 0.260E-08 |
| 0.10000 | 0.300E-08 |
| 0.12000 | 0.325E-08 |
| 0.14000 | 0.334E-08 |
| 0.16000 | 0.326E-08 |
| 0.18000 | 0.301E-08 |
| 0.20000 | 0.261E-08 |
| 0.22000 | 0.206E-08 |
| 0.24000 | 0.143E-08 |
| 0.26000 | 0.704E-09 |
| 0.28000 | -0.616E-10 |
| 0.30000 | -0.828E-09 |
| 0.32000 | -0.156E-08 |
| 0.34000 | -0.220E-08 |
| 0.36000 | -0.273E-08 |
| 0.38000 | -0.311E-08 |
| 0.40000 | -0.331E-08 |
| 0.42000 | -0.332E-08 |
| 0.44000 | -0.313E-08 |
| 0.46000 | -0.275E-08 |
| 0.48000 | -0.219E-08 |
| 0.50000 | -0.149E-08 |
| 0.52000 | -0.678E-09 |
| 0.54000 | 0.188E-09 |
| 0.56000 | 0.105E-08 |
| 0.58000 | 0.186E-08 |
| 0.60000 | 0.254E-08 |
| 0.62000 | 0.304E-08 |
| 0.64000 | 0.330E-08 |
| 0.66000 | 0.330E-08 |
| 0.68000 | 0.302E-08 |
| 0.70000 | 0.245E-08 |
| 0.72000 | 0.163E-08 |
| 0.74000 | 0.631E-09 |
| 0.76000 | -0.472E-09 |
| 0.78000 | -0.156E-08 |
| 0.80000 | -0.249E-08 |
| 0.82000 | -0.313E-08 |
| 0.84000 | -0.334E-08 |
| 0.86000 | -0.300E-08 |
| 0.88000 | -0.208E-08 |
| 0.90000 | -0.641E-09 |
| 0.92000 | 0.109E-08 |
| 0.94000 | 0.267E-08 |
| 0.96000 | 0.333E-08 |
| 0.98000 | 0.190E-08 |
| 1.00000 | -0.334E-08 |




MAX ERROR= 0.33379E-08

第3表 FACOM 230-60 による奇関数の最良近似有理式を求めるプログラム

```
      OPTION  LIST,DOUBLE
      COMMON L,M,N,P(40),Q(40),Z(40,41)
      COMMON/PARAM/A,B,EPS1,EPS2
      DATA HPI/1.5707963267948966192/
C FUNCTION
      F(X)= SIN(HPI*X)
C WEIGHT FUNCTION
      W(X)= 1.0+(X-X)
C ERROR FUNCTION
      E(X)= W(X)*(F(X)-PQ(X*X)*X)
      A=-1.0
      B=+1.0
      EPS1=EPS2=1.0E-4
      DO 10 N=1,4
      DO 10 L=0,N
      M= N-L
   10 CALL KIGU(F,E,W,*3)
    3 FORMAT(1H1///10X,'SIN(HPI*X)'//)
      STOP
      END

      SUBROUTINE KIGU(F,E,W,FMT)
      COMMON L,M,N,P(40),Q(40),Z(40,41)
      COMMON/PARAM/A,B,EPS1,EPS2
      DIMENSION FMT(1),X(101),EX(101)
C
      PN= 1.5707963/FLOAT(2*L+2*M+3)
      A1=0.5*(B-A)
      A2= 0.5*(B+A)
C (A)
      DO 7 I=1,L+1
    7 P(I)= 0.0
      DO 8 I=1,M+1
    8 Q(I)= 0.0
      Q(1)= 1.0
      DO 9 I=1,N+1
    9 X(I)=A1*SIN(FLOAT(2*I-1)*PN) + A2
      X(N+2)= B
C (B)
      DO 20 I=1,N+1
      U =A1*SIN(FLOAT(2*I)*PN) + A2
      FU= F(U)
      U2= U*U
      Z(I,1)= U
      IF(L.EQ.0) GO TO 21
      DO 22 J=1,L
   22 Z(I,J+1)= Z(I,J)*U2
   21 IF(M.EQ.0) GO TO 23
      Z(I,L+2)= -FU*U2
      IF(M.EQ.1) GO TO 23
      DO 24 J= 2,M
   24 Z(I,L+J+1)= Z(I,L+J)*U2
   23 Z(I,N+2)= FU
   20 CONTINUE
C (C)
   99 DO 31 K=1,N+1
      DO 32 I=K+1,N+2
   32 Z(K,I)= Z(K,I)/Z(K,K)
      DO 33 I=1,N+1
      IF(K-I) 34,33,34
   34 DO 36 J=K+1,N+2
   36 Z(I,J)= Z(I,J)-Z(I,K)*Z(K,J)
   33 CONTINUE
   31 CONTINUE
C (D)
      IF(M.EQ.0) GO TO 40
      DO 41 I=1,M
   41 Q(I+1)= Q(I+1)+Z(L+I+1,N+2)
   40 DO 42 I=1,L+1
   42 P(I)= P(I) + Z(I,N+2)
C (E)
      XA= 0.0
      XB= 0.5*(X(1)+X(2))
      CALL YAMA(E,EPS1,X(1),XA,XB)
      DO 43 I=1,N
      XA= 0.5*(X(I)+X(I+1))
      XB= 0.5*(X(I+1)+X(I+2))
   43 CALL YAMA(E,EPS1,X(I+1),XA,XB)
C (F)
      EMAX= 0.0
      EMIN= 1.0E+70
      DO 51 I=1,N+2
      EE=EX(I)=E(X(I))
      IF(ABS(EE).GT.EMAX) EMAX=ABS(EE)
   51 IF(ABS(EE).LT.EMIN) EMIN=ABS(EE)
      IF(EMAX-EMIN.LT.EPS2*EMAX) GO TO 90
C (G)
      U= X(1)
      U2= U*U
      FP= POL(P,L,U2)*U
      FQ= POL(Q,M,U2)
      WU= W(U)
      Z(1,1)=(WU*U)/FQ
      IF(L.EQ.0) GO TO 81
      DO 82 J=1,L
   82 Z(1,J+1)= Z(1,J)*U2
   81 IF(M.EQ.0) GO TO 83
      Z(1,L+2)= -WU*FP*U2/(FQ*FQ)
      IF(M.EQ.1) GO TO 83
      DO 84 J=2,M
   84 Z(1,L+J+1)= Z(1,L+J)*U2
   83 Z(1,N+2) = E(U)
      DO 85 I=1,N+1
      U = X(I+1)
      U2= U*U
      FP= POL(P,L,U2)*U
      FQ= POL(Q,M,U2)
      WU= W(U)
      Z(I+1,1)=(WU*U)/FQ
      IF(L.EQ.0) GO TO 86
      DO 87 J=1,L
   87 Z(I+1,J+1)= Z(I+1,J)*U2
   86 IF(M.EQ.0) GO TO 88
      Z(I+1,L+2)=-WU*FP*U2/(FQ*FQ)
      IF(M.EQ.1) GO TO 88
      DO 89 J= 2,M
   89 Z(I+1,L+J+1)= Z(I+1,L+J)*U2
   88 Z(I+1,N+2)= E(U)
      DO 95 J=1,N+2
   95 Z(I,J)= Z(I,J)+Z(I+1,J)
   85 CONTINUE
      GO TO 99
C (H)
   90 WRITE(6,FMT)
      DO 91 I=1,L+1
      K= 2*I-1
   91 WRITE(6,101) K,P(I)
  101 FORMAT(5X,2HP(,I3,2H)=,E30.18)
      WRITE(6,102)
  102 FORMAT(1X)
      DO 92 I=1,M+1
      K= 2*I-2
   92 WRITE(6,103) K,Q(I)
  103 FORMAT(5X,2HQ(,I3,2H)=,E30.18)
      WRITE(6,201)( I,X(I),EX(I),I=1,N+2)
  201 FORMAT(/(I12,2E20.8))
      DO 60 I=1,101
      X(I)= A+(B-A)*FLOAT(I-1)/100.0
   60 FX(I)=E(X(I))
      CALL CURVE(101,X,FX,EMAX)
      RETURN
      END
```

第4表 sin の結果の例

```
      SUBROUTINE  YAMA(F,EPS,X,XA,XB)
      KEY=9
      H=(XB-XA)*0.25
      F0=ABS(F(X))
   99 CONTINUE
      X1=X+H
      IF(X1.GT.XB) X1=XB
      X2=X-H
      IF(X2.LT.XA) X2=XA
      F1=ABS(F(X1))
      F2=ABS(F(X2))
      IF(F2.LE.F0.AND.F0.GE.F1) GO TO 10
      IF(F2-F1) 1,10,2
    1 X=X1
      F0=F1
      IF(KEY.EQ.2) H=0.5*H
      KEY=1
      GO TO 99
    2 X=X2
      F0=F2
      IF(KEY.EQ.1) H=0.5*H
      KEY=2
      GO TO 99
   10 H=0.5*H
      IF(ABS(F0-F1)+ABS(F0-F2)
     C      .GE. EPS*F0) GO TO 99
      RETURN
      END
```

```
      SUBROUTINE  CURVF(N,X,E,AMAX)
      DIMENSION IA(103),IB(103),IC(11)
      DIMENSION X(101),E(101)
      DATA IA/103*1H /,IB/103*1H-/
      DATA   IK,II,IS/1H*,1HI,1H  /
      DATA IC/-50,-40,-30,-20,-10,
     C             0,10,20,30,40,50/
      WRITE(6,20)
   20 FORMAT(1H1//)
      WRITE(6,40) IC
   40 FORMAT(22X,5I10,I9,5I10)
      WRITE(6,50) II
   50 FORMAT(30X,10(1HI,9X),A1)
      WRITE(6,70) IB
   70 FORMAT(1H ,5X,1HX,14X,1HE,7X,103A1)
      DO 100 I=1,N
      IV=50.0*E(I)/AMAX+52.5
      IF(IV.LT.1  ) IV=1
      IF(IV.GE.103) IV=103
      IA(1)=IA(52)=IA(103)=II
      IA(IV)=IK
      WRITE(6,200)X(I),E(I),IA
  100 IA(IV)= IS
  200 FORMAT(1H ,F10.5,3X,E13.3,2X,103A1)
      WRITE(6,70) IB
      WRITE(6,50) II
      WRITE(6,40) IC
      WRITE(6,300)AMAX
  300 FORMAT(1H  ,10HMAX ERROR=  ,E15.5)
      RETURN
      END
```

```
      FUNCTION  POL(A,N,X)
      DIMENSION  A(N)
      POL=A(N+1)
      IF(N.EQ.0)RETURN
      DO 10 I=N,1,-1
   10 POL=POL*X+A(I)
      RETURN
      END
```

```
      FUNCTION  PQ(X)
      COMMON  L,M,N,P(40),Q(40),Z(40,41)
      PX = P(L+1)
      IF(L.EQ.0) GO TO 15
      DO 10 I=1,L
   10 PX = PX*X + P(L+1-I)
   15 QX= Q(M+1)
      IF(M.EQ.0) GO TO 25
      DO 20 I=1,M
   20 QX = QX*X + Q(M+1-I)
   25 PQ = PX/QX
      RETURN
      END
```

第4表 sin の結果の例

$$\sin\left(\frac{\pi}{2}x\right), \qquad -1.0 \leqq x \leqq 1.0$$

```
 (次数)                (係数)

P(  1)=      0.15707962896028l881E 01
P(  3)=     -0.51586909802295662 8E 00
P(  5)=      0.30874054731024709E-01

Q(  0)=      0.10000000000000000E 01
Q(  2)=      0.82820564711789554 5E-01
Q(  4)=      0.29856779748650410 1E-02

      1      0.14068388E 00       0.33378260E-08
      2      0.41090369E 00      -0.33378260E-08
      3      0.64983024E 00       0.33378260E-08
      4      0.83808776E 00      -0.33378260E-08
      5      0.95850724E 00       0.33378260E-08
      6      0.10000000E 01      -0.33378260E-08

             (極大点)             (最大誤差)
```

## 参考文献


1) 山下："有理式の最良近似式を求めるプログラム", 情報処理, Vol. 10, No. 6. pp. 304 (1969)

2) 山下："偶関数の最良近似有理式を求めるプログラム", 情報処理, Vol. 12, No. 6, pp. 372 (1971).


(昭和46年4月16日受付)